main:

   call InitSensors(0,0,0,0)

   call ReadGPS(Long,Lat)

   xyString 10,100,"Longitude: ",Long;" "

   xyString 10,120,"Latitude:  ",Lat;" "

end

sub ReadGPS(&a,&b)

   // give the Interface Utility a command

   setcbtext("GPS") 

   // now request the Utility to perform 

   filerename ("RBsTurn.txt","SensorsTurn.txt")

   // and wait for it to respond

   while !fileexists("RBsTurn.txt") 

   wend

   // then get the data from the clipboard

   s=getcbtext()

   // and parse it

   mfromstring Sensors,s,"," 

   // and return the values as variables

   b=Sensors[1]

   a=Sensors[2]

   // round to 5 decimal places

   a= Round(a*100000)/100000.0

   b= Round(b*100000)/100000.0

return  

Figure 2

sub GetImage(Command,type)  

  FileDelete("Temp.jpg")

  // tell the Utility to take the picture

  SetCBtext(Command) 

  FileRename ("RBsTurn.txt","SensorsTurn.txt")  

  If type="BMP" then FileDelete("Temp.bmp")

  // wait for handshake

  while !FileExists("Temp.jpg")

  wend

  delay 100

  if type="BMP" then ToBmp "Temp.jpg"

return
Figure 4

main:

  Dim CntInfo[9,9]

  BMP="BMP"

  JPG="JPG"

  dx=800/9 // scrn resolution is 800x600

  dy=600/9

  // wait 3 seconds then take picture

  delay 3000

  call GetImage("Cam1 10",BMP)

  // display image

  FitBmp "Temp"

  // show grid for clarity

  for i=0 to 800 step dx 

    line i,0,i,600,1,LightBlue

  next

  for i=0 to 600 step dy

    line 0,i,800,i,1,LightBlue

  next

  xyString 230,550, "Touch the object to be found"

  repeat

    ReadMouse x,y,b

  until b

  // record objects color in col

  ReadPixel x,y,col

  // track object continuously 

  while true

    call GetImage("Cam1 10",BMP)

    // display image

    Found=-1

    FitBmp "Temp"

    BMPfindClr "Temp",col,Found,SecCnt,.2,.05,9,CntInfo

    if Found>-1

      // determine grid coordinates of found block

      x=Found#9

      y=Found/9

      // outline the block in red

      line x*dx,y*dy,(x+1)*dx,y*dy,4,red

      lineto (x+1)*dx,(y+1)*dy,4,red

      lineto x*dx,(y+1)*dy,4,red

      lineto x*dx,y*dy,4,red 

    endif

  wend

end
Figure 5

main:

  #include "RROScommands.bas"

  gosub InitCommands

  call InitSensors(0,0,0,180)

  gosub InitRB9robot // initialize the robot

  Dim CntInfo[9,9]

  BMP="BMP"

  //dx=800/9

  //dy=600/9

  // wait 3 seconds then take picture

  delay 000

  for i=1 to 20 // allow auto adjust

    call GetImage("Cam1 10",BMP)

    // display image

    FitBmp "Temp"

  next

  xyString 230,550, "Touch the object to be found"

  repeat

    ReadMouse x,y,b

  until b

  ReadPixel x,y,col // place color at x,y in variable col

  DestLong = -80.45905 // manually set destination

  DestLat = 27.66943

  ClearScr

  xyString 100,300, "Move the robot to a remote location and touch the screen."

  repeat // wait till mouse button released

    ReadMouse x,y,b

  until !b

  repeat  // wait till mouse button pressed

    ReadMouse x,y,b

  until b

  // use GPS and compass to move to previous location

  LastLong = DestLong

  LastLat =  DestLat

  Done=0

  DirToMove=0

  repeat

     Call ReadGPS(CurLong,CurLat)

     if (CurLong = LastLong) and (CurLat = LastLat)

       // keep moving at DirToMove

       call Orientation(T1,T2,T3,Heading)

       call AngleDifference(Heading,DirToMove,dif)

       if abs(dif)<6 // move forward if on course

         rForward 1

       elseif dif<0

         rCommand(SetTurnStyle,70)

         rTurn –1

       elseif dif>0

         rCommand(SetTurnStyle,70)

         rTurn 1

       endif

     else

       // calculate a new CurAngle (Direction to Move)

       call Orientation(T1,T2,T3,Heading)

       call Angle(CurLong,CurLat,DestLong,DestLat,DirToMove,done)

       call AngleDifference(Heading,DirToMove,dif)

       rTurn dif

       LastLong=CurLong

       LastLat=CurLat

     endif

  until done

  // use camera to find the object

  move=0

  cnt=0

  repeat

    call GetImage("Cam1 10",BMP)

    Found=-1

    BMPfindClr "Temp",col,Found,SecCnt,.2,.05,9,CntInfo

    if Found>-1 // if the object is seen

      // determine grid coordinates of block

      x=Found#9

      y=Found/9

      // and move toward object

      if x<4

        rCommand(SetTurnStyle,80)

        rTurn –1

      elseif x>4

        rCommand(SetTurnStyle,80)

        rTurn 1

      else

        rForward 1

      endif

    else // object not seen – look for it

      if move=0 

        rTurn –30 // look left

      elseif move=1 

        rTurn 60  // look right

      elseif move=2

        rTurn –30 // back to center

        rForward 100 // forward a little

        move=0 // start over

        cnt++ // increment type of move to make

      endif 

      move++ // increment # of tries   

    endif

  until ((Found>-1)and(x=4)and(rRange()<30))or(cnt=5)

  if cnt=5 

    // 5 tries, no object seen

    print "Object not found – I give up"  

  else

    print "Object Found"

  endif   

end

sub Angle(CurLon,CurLat,DestLon,DestLat,&an,&here)

  CurLon=abs(CurLon)

  DestLon=abs(DestLon)

  lon=(CurLon-DestLon) // delta x                    

  lat=(CurLat-DestLat) // delta y

  here=false

  if lon!=0 then a=rtod(atan(abs(lat/lon)))

  if lon=0 and lat=0 then an=0\here=true

  if lon=0 and lat<0 then an=0

  if lon=0 and lat>0 then an=180

  if lon<0 and lat<=0 then an=270+a

  if lon>0 and lat>=0 then an=90+a

  if lon>0 and lat<=0 then an=90-a

  if lon<0 and lat>=0 then an=270-a

return

// routines not shown are listed in Fig 4 & 5

// or in previous articles in this series 

